
1)  Overview

This document lists all the changes needed to correct problems in the TTCN implementation of test case 9.2.3.4.1, which is part of the LTE test suite. The test case can be demonstrated to run with an LTE UE (see section 6). Execution log files are provided as evidence.

3) Verification Test Summary

Test Case:
9.2.3.4.1
Test Group:
EMM
System Simulator used:
Anite Conformance Toolset
UE used:
Nvidia ICERA 410
Verification Status:
PASS

4) Corrections required for test case 9.2.3.4.1

4.1 Introduction

This section describes the changes required to make test case 9.2.3.4.1 run correctly with a LTE UE. The TTCN used to pass this test case belongs to iwd-EUTRA-B2011-03_D12wk09. Changes added through R5s120119 are also applicable to this submission.
4.2 Change 1

	Testcase name
	9.2.3.4.1

	Reason for change
	Initial registration on LTE cell should be done as per the MCC and MNC indicated in initial condition. However in the TC body GERAN and LTE cell should be with default MCC and MNC.

Note: Prose CR will be raised for these changes at the next RAN5 meeting

	Summary of change
	Initially we created cell B with these MNC and MCC and later in TC body we switch off cell B and use cell A with default MMC and MCC.

	Source of change
	NAS_TrackingArea_UG.ttcn


Before change:

	  function f_TC_9_2_3_4_1_EUTRA() runs on EUTRA_PTC

  { /* @desc TAU/RAU procedure for inter-system cell re-selection between A/Gb and S1 modes */

    var template (value) PLMN_Identity v_HPLMN_987_65 := { mcc := {9, 8, 7},  mnc := {6, 5} };

    var  template(value) GutiParameters_Type v_Guti_Parameters := cs_Guti(v_HPLMN_987_65, 65244, 186);

    f_EUTRA_NAS_Init(LTE_AllCellsOnSamePLMN, c5);

    f_EUTRA_CellInfo_SetGuti(eutra_CellA, v_Guti_Parameters);

    // create cell and send out system information

    f_EUTRA_CellConfig_Def(eutra_CellA);

    // Preamble: The UE is in state Switched OFF (state 1)

    // ==========================================================================

    f_EUTRA_Preamble_InterRAT(eutra_CellA, STATE2_IDLEUPDATE, GERAN);

    f_EUTRA_UE_Detach_SwitchOffUe ( eutra_CellA, RRC_IDLE );

    // switch off cell A.

    f_EUTRA_SetCellPower(eutra_CellA, tsc_NonSuitableOffCellRS_EPRE);

    f_EUTRA_TestBody_Set(true);

    fl_TC_9_2_3_4_1_Body();

    // Postamble

    f_EUTRA_TestBody_Set(false);

    f_EUTRA_Postamble(eutra_CellA, E2_T3440); // @sic R5s110007 sic@

  } // function f_TC_9_2_3_4_1_EUTRA


After change:

	  function f_TC_9_2_3_4_1_EUTRA() runs on EUTRA_PTC

  { /* @desc TAU/RAU procedure for inter-system cell re-selection between A/Gb and S1 modes */

    var template (value) PLMN_Identity v_HPLMN_987_65 := { mcc := {9, 8, 7},  mnc := {6, 5} };

    var  template(value) GutiParameters_Type v_Guti_Parameters := cs_Guti(v_HPLMN_987_65, 65244, 186);

    f_EUTRA_NAS_Init(LTE_AllCellsOnSamePLMN, c5);

    f_EUTRA_CellInfo_SetGuti(eutra_CellB, v_Guti_Parameters);

    // create cell and send out system information

    f_EUTRA_CellConfig_Def(eutra_CellA);

    f_EUTRA_CellConfig_Def(eutra_CellB);
    // Preamble: The UE is in state Switched OFF (state 1)

    // ==========================================================================

    f_EUTRA_Preamble_InterRAT(eutra_CellB, STATE2_IDLEUPDATE, GERAN);

    f_EUTRA_UE_Detach_SwitchOffUe ( eutra_CellB, RRC_IDLE );

    // switch off cell A.

    f_EUTRA_SetCellPower(eutra_CellB, tsc_NonSuitableOffCellRS_EPRE);

    f_EUTRA_TestBody_Set(true);

    fl_TC_9_2_3_4_1_Body();

    // Postamble

    f_EUTRA_TestBody_Set(false);

    f_EUTRA_Postamble(eutra_CellA, E2_T3440); // @sic R5s110007 sic@

  } // function f_TC_9_2_3_4_1_EUTRA


4.3 Change 2

	Testcase name
	9.2.3.4.1

	Reason for change
	1.   At step 17, PLMN ID should not be checked but currently its presence is checked. Also Mmegi and Mmec should be drived from LAC and P-TMSI sent at GERAN side at test step 8 as per the test spec. Currently Mmegi and Mmec are drived from v_Guti1_Params which is assigned during preamble.

Note: most significant bit of MME_GroupId should be replaced with 0 where it is used as LAC received from GERAN. Like while creating v_GutiStringFromGERAN from MME_GroupId, most significant bit should be replace with 0

2.   At step 17 & 28 Old PTMSI Signature & Nonce will be sent by UE in TAU request message.

(Spec Reference 24.008 Clause 4.7.3.1.3, 24.031 Clause 8.2.29.4,8.2.29.6)

3GPP TS 24008

4.7.3.1.3
GPRS attach accepted by the network

Upon receiving the ATTACH ACCEPT message an MS supporting S1 mode shall set the TIN to "P-TMSI".

3GPP TS 24301

8.2.29.4
Old P-TMSI signature

The UE shall include this IE if the TIN indicates "P-TMSI" and the UE holds a valid P-TMSI signature, P-TMSI and RAI.

8.2.29.6
NonceUE

This IE is included if the UE performs an A/Gb mode or Iu mode to S1 mode inter-system change in idle mode

3.   At step 22 the EUTRAN cell should be made Non Suitable for cell reselection to GERAN however in TTCN the EUTRA was made suitable during cell reslection to GERAN due to which UE wont move to EUTRA.

4.   At step 22 AuthCord Parameters are not sent to GERAN side and also v_Guti2_Params is not set in cell parameters. Currently TTCN is calling f_SendIRATCoOrd to send the trigger to GERAN side in place of f_EUTRA_SendAuthCoOrdParameters;

5.   At step 27, f_ReceiveAuthCoOrdParameters will be used to receive AuthCordParams however TTCN is using f_WaitForIRATCoOrd_Trigger

6.   At step 28, RegisteredMME will not contain selected PLMN ID as per the test spec clause “Table 9.2.3.4.1.3.3-7: Message RRCConnectionSetupComplete (step 28, Table 9.2.3.4.1.3.2-1)”. Also Mmegi and Mmec will be drived from MME_GroupId and MME_Code (mapped from LAC sent at GERAN side at test step 25 and P-TMSI sent at GERAN side at test step 8 respectively) rather than are from v_Guti2_Params 

According to 24.301 5.3.1.1

5.3.1.1
Establishment of the NAS signalling connection

For the routing of the initial NAS message to the appropriate MME, the UE NAS provides the lower layers with either the S-TMSI or the registered globally unique MME identifier (GUMMEI) that consists of the PLMN ID, the MME group ID, and the MME code (see 3GPP TS 23.003 [2]) according to the following rules:

-
When the UE is registered in the tracking area of the current cell during the NAS signalling connection establishment, the UE NAS shall provide the lower layers with the S-TMSI, but shall not provide the registered MME identifier to the lower layers. Exceptionally, when the UE in EMM-IDLE mode initiates a tracking area updating or combined tracking area updating procedure for load balancing purposes, the UE NAS shall provide the lower layers with neither S-TMSI nor registered MME identifier.

-
When the UE is not registered in the tracking area of the current cell during the NAS signalling connection establishment, the UE NAS does not provide the lower layers with the S-TMSI. Instead,

a)
if the TIN indicates "GUTI" or "RAT-related TMSI", or the TIN is not available, and the UE holds a valid GUTI, the UE NAS shall provide the lower layers with the MME identifier part of the valid GUTI; or

b)
if the TIN indicates "P-TMSI" and the UE holds a valid P-TMSI and RAI, the UE NAS shall provide the lower layers with the MME identifier part of the mapped GUTI, which is generated from the P-TMSI and RAI.

Note: most significant bit of MME_GroupId should be replaced with 0 where it is used as LAC received from GERAN.

7.   At step 28 the update type for TAU Request will be “combined TA/LA updating with IMSI attach” when requested TAU type as “Combined TaLaUpdate” howver TTCN contains “combined TA/LA updating only”

8.   At step 30 SRB1 will be used in TAUComplete however TTCN was using SRB2

9.   After step 30 the v_Guti3_Params will be set before the postamble or it will create mismatch in GUTI of DetachRequest.

Note: Prose CR will be raised for these changes at the next RAN5 meeting

	Summary of change
	1.   At test step 17, following changes are done

a) While creating v_GutiStringFromGERAN from MME_GroupId, most significant is replaced with 0

b) PLMN ID received inside template cr_RegisteredMME is changed from ‘?’ to ‘*’

c) Mmegi and Mmec are drived from MMEGI Group and MME code received from GERAN side from v_CoOrd_AuthParams

2.   New template cdr_TAU_Request_OldandAdditionalGuti_92341 is defined and used in Step 17 & 25 to receive Naunce and Old PTMSI Signature.

3.   Cell A made non suitable after test step 20

4.   v_Guti2_Params is set in cell information and also  f_EUTRA_SendAuthCoOrdParameters is called in place of f_SendIRATCoOrd to pass the auth parameter to GERAN side

5.   At step 27, f_ReceiveAuthCoOrdParameters is used to receive AuthCordParams instead of f_WaitForIRATCoOrd_Trigger

6.   At test step 28, following changes are done for reason 8

a.
PLMN ID received inside template cr_RegisteredMME is changed from omit

b.
Mmegi and Mmec are drived from MMEGI Group and MME code received from GERAN side from v_CoOrd_AuthParams (NOTE: At GERAN side, MMEGI Group is drived from LAC sent at step 25 and MME Code drived from PTMSI sent at step 8 at GERAN side before sending the parameters to EUTRA side)

7.   At step 28 the update type for TAU Request is made combined TA/LA updating with IMSI attach for teh case of “Combined TaLaUpdate” At step 30 SRB1 is used in TAU Complete

8.   At step 30 SRB2 used in TAUComplete replaced with SRB1

9.   At step 30 v_Guti3_Params is set before the postamble

	Source of change
	NAS_TrackingArea_UG.ttcn


Before change:

	  function fl_TC_9_2_3_4_1_Body() runs on EUTRA_PTC

  {

    var SRB_COMMON_IND v_ReceivedAsp;

    var template (omit) EPS_BearerContextStatus v_EpsBearerCtxtStatus;

    var IRAT_CoOrd_Auth_Type v_CoOrd_AuthParams;

    var EPS_UpdateTypeValue v_UpdateType := f_GetEPSTAUType(NORMAL); // @sic R5s110176 sic@

    // ==========================================================================

    // initialize component specific data structures

    // ==========================================================================

    // get all data which are required for a GUTI

    var GutiParameters_Type v_Guti1_Params := f_EUTRA_CellInfo_GetGuti(eutra_CellA);

    var GutiParameters_Type v_Guti2_Params := f_CreateGutiParamsWithUpdatedSTmsi(v_Guti1_Params);

    var GutiParameters_Type v_Guti3_Params := f_CreateGutiParamsWithUpdatedSTmsi(v_Guti2_Params);

    var O10_Type v_GutiStringFromGERAN;

    var NAS_PlmnId v_PLMN := f_Asn2Nas_PlmnId(v_Guti1_Params.PLMN_Identity);

    var NAS_Tac v_TAC := bit2oct(f_EUTRA_CellInfo_GetTAC(eutra_CellA));

    var RRC_TransactionIdentifier v_RRC_TI := tsc_RRC_TI_Def;

    var NAS_Lac v_LAC := f_EUTRA_CellInfo_GetLocationAreaCode ( eutra_CellA );

    var template (omit) LocAreaId v_LAI := f_GetLAI ( v_PLMN, v_LAC, NORMAL); // @sic R5s110176 sic@

    var EUTRA_SecurityParams_Type v_SecurityParams := f_EUTRA_Security_Get();

    var Common_AuthenticationParams_Type v_Auth_Params;

    var template AdditionalUpdateType v_AdditionalUpdateType := f_GetAdditionalUpdateType ();

    v_SecurityParams.AuthParams.KeySeq := v_SecurityParams.KSIasme; // The KSIasme value needs to be passed over as it will be used as the GERAN CKSN

    f_EUTRA_Security_Set(v_SecurityParams);

    v_Auth_Params := f_AuthenticationInit(v_SecurityParams.AuthParams);

    f_EUTRA_SendAuthCoOrdParameters(GERAN, eutra_CellA, v_Auth_Params);

    //+ Steps 1 to step 9 are implemented in GERAN side.

    //@siclog "Step 10 - 12 Void" siclog@

    //@siclog "Step 13" siclog@

    //+   The signal strength of Cell A is raised to that of the Serving Cell and that of Cell 24 is lowered to that of a Suitable Neighbour Cell as defined in table 6.2.2.1-1 of TS 36.508.

    //    Note: Cell 24 is still suitable but the UE is expected to select Cell A.

    v_CoOrd_AuthParams := f_ReceiveAuthCoOrdParameters(GERAN);

    f_EUTRA_SetCellPower(eutra_CellA, tsc_ServingCellRS_EPRE);

    v_GutiStringFromGERAN := v_PLMN & bit2oct(v_CoOrd_AuthParams.GutiParams.MME_GroupId) & bit2oct(v_CoOrd_AuthParams.GutiParams.MME_Code) & bit2oct(v_CoOrd_AuthParams.GutiParams.M_TMSI); 

    //@siclog "Step 14 Void" siclog@

    //@siclog "Step 15" siclog@

    //+   Check: does the UE send an RRCConnectionRequest with the InitialUE-Identity set to "randomValue" and the establishmentcause

    //     set to MO-signalling on Cell A?

    // Editor's Note: Cell reselection to E-UTRAN might not occur until the GPRS READY timer has expired. Hence this step might occur up to 40 seconds after step 13.

    f_EUTRA_RRC_ConnectionRequest_Def(eutra_CellA, cr_EstablishmentCause_MoSignalling, cr_InitialUE_Identity_NotSTMSI);   /* @sic R5-120721 change 3 sic@ */

    //@siclog "Step 16" siclog@

    //+   The SS responds with RRCConnectionSetup

    f_EUTRA_RRC_ConnectionSetup_Def ( eutra_CellA );

    //@siclog "Step 17" siclog@

    //+   Check: does the UE send an RRCConnectionSetupComplete with the mmegi and mmec set to the values derived from the mapped RAI and P-TMSI; [1,P]

    //  Check: are the contents of the TRACKING AREA UPDATE REQUEST with the correct parameters? [2,3,P]

    SRB.receive(car_SRB1_RrcNasPdu_IND(eutra_CellA,

                                       cr_508_RRCConnectionSetupComplete(v_RRC_TI, ?,

                                                                         cr_RegisteredMME(?,

                                                                                          v_Guti1_Params.MME_GroupId,

                                                                                          v_Guti1_Params.MME_Code)),

                                       cr_NAS_Indication(tsc_SHT_IntegrityProtected,

                                                         cdr_TAU_Request_OldandAdditionalGuti(v_UpdateType,

                                                                                              f_EUTRA_SecurityKSIasme_Get(),

                                                                                              v_AdditionalUpdateType,

                                                                                              cs_MobileIdentityGuti(omit, v_GutiStringFromGERAN),

                                                                                              cr_CiphKeySeqNum ('8'H, v_CoOrd_AuthParams.AuthPS.KeySeq),

                                                                                              f_GutiParameters2MobileIdentity(tsc_IEI_Guti, v_Guti1_Params),

                                                                                              cr_UeRadioCap)))) -> value v_ReceivedAsp;

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 17");

    //@siclog "Step 18" siclog@

    //+   The SS sends TRACKING AREA UPDATE ACCEPT

    //  Note: the default message contents cause the allocation of a new GUTI and new TAI list

    v_EpsBearerCtxtStatus := v_ReceivedAsp.Signalling.Nas[0].Pdu.Msg.tRACKING_AREA_UPDATE_REQUEST.epsBearerContextStatus;

    SRB.send(cas_SRB1_NasPdu_REQ(eutra_CellA,

                                 cs_TimingInfo_Now,

                                 cs_NAS_Request(

                                                tsc_SHT_IntegrityProtected_Ciphered,

                                                cs_508_TAU_Accept(

                                                                  v_UpdateType,

                                                                  f_GutiParameters2MobileIdentity(tsc_IEI_Guti, v_Guti2_Params),

                                                                  cds_TAIListNonConsecutive_tlv(v_PLMN,

                                                                                                { '0001'O, '0002'O }), // TAC1=1, TAC2=2

                                                                  v_EpsBearerCtxtStatus,

                                                                  v_LAI,

                                                                  f_GetMSId (NORMAL), // @sic R5s110176 sic@

                                                                  f_GetAdditionalUpdateResult(v_ReceivedAsp.Signalling.Nas[0].Pdu.Msg.tRACKING_AREA_UPDATE_REQUEST.addUpdateType)) ) ) );

    //+  The UE sends TRACKING AREA UPDATE COMPLETE

    //@siclog "Step 19" siclog@

    SRB.receive(car_SRB1_NasPdu_IND(eutra_CellA,

                                    cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                      cr_508_TAU_Complete)));

    //+ the SS releases the RRC connection

    //@siclog "Step 20" siclog@

    f_EUTRA_RRC_ConnectionRelease(eutra_CellA);

    //+ Cell 24 is switched off

    //@siclog "Step 21" siclog@

    //@siclog "Step 22" siclog@

    // The signal strength of Cell 26 is raised to that of the Serving Cell and that of Cell A is lowered to that of a Suitable Neighbour Cell as defined in table 6.2.2.1-1 of TS 36.508.

    // Note: Cell A is still suitable but the UE is expected to select Cell 26.

    f_EUTRA_SetCellPower(eutra_CellA, tsc_SuitableCellRS_EPRE);

    f_SendIRATCoOrd (GERAN, cms_IRAT_Trigger);
    //+ Step 21 to step 26 is implemented in GERAN side

    //@siclog "Step 27" siclog@

    //+  The signal strength of Cell A is raised to that of the Serving Cell and that of Cell 26 is lowered to that of a Suitable Neighbour Cell as defined in table 6.2.2.1-1 of TS 36.508.

    //   Note: Cell 26 is still suitable but the UE is expected to select Cell A.

    f_WaitForIRATCoOrd_Trigger(GERAN);
    f_EUTRA_SetCellPower(eutra_CellA, tsc_ServingCellRS_EPRE);

    //@siclog "Step 28" siclog@

    // Check: does the UE send on Cell A an RRCConnectionSetupComplete with the selectedPLMN-identity indicating the registered MME

    // (i.e. the PLMN in the GUTI allocated in step 18 ), and, the mmegi and mmec are set to the values in the GUTI allocated in step 18? [1,P]

    // Check: does the UE send a TRACKING AREA UPDATE REQUEST with the correct parameters? [2,6,P]

    f_EUTRA_RRC_ConnectionRequest_Def(eutra_CellA);    // @sic R5-120721 change 3 sic@

    f_EUTRA_RRC_ConnectionSetup_Def(eutra_CellA);

    SRB.receive(car_SRB1_RrcNasPdu_IND(eutra_CellA,

                                       cr_508_RRCConnectionSetupComplete(v_RRC_TI, ?,

                                                                         cr_RegisteredMME(v_Guti2_Params.PLMN_Identity,

                                                                                          v_Guti2_Params.MME_GroupId,
                                                                                          v_Guti2_Params.MME_Code)),

                                       cr_NAS_Indication(tsc_SHT_IntegrityProtected,

                                                         cdr_TAU_Request_OldandAdditionalGuti(v_UpdateType,

                                                                                              f_EUTRA_SecurityKSIasme_Get(),

                                                                                              v_AdditionalUpdateType,

                                                                                              cs_MobileIdentityGuti(omit, v_GutiStringFromGERAN),

                                                                                              *,

                                                                                              f_GutiParameters2MobileIdentity(tsc_IEI_Guti, v_Guti2_Params),

                                                                                              omit)))) -> value v_ReceivedAsp;

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 28");

    //@siclog "Step 29" siclog@

    // The SS sends TRACKING AREA UPDATE ACCEPT

    // Note: the default message contents cause the allocation of a new GUTI.

    v_EpsBearerCtxtStatus := v_ReceivedAsp.Signalling.Nas[0].Pdu.Msg.tRACKING_AREA_UPDATE_REQUEST.epsBearerContextStatus;

    SRB.send ( cas_SRB1_NasPdu_REQ ( eutra_CellA,

                                     cs_TimingInfo_Now,

                                     cs_NAS_Request ( tsc_SHT_IntegrityProtected_Ciphered,

                                                      cs_508_TAU_Accept ( v_UpdateType,

                                                                          f_GutiParameters2MobileIdentity(tsc_IEI_Guti, v_Guti3_Params),

                                                                          cds_TAIListNonConsecutive_tlv( v_PLMN, {v_TAC } ),

                                                                          v_EpsBearerCtxtStatus,

                                                                          v_LAI,

                                                                          f_GetMSId (NORMAL), // @sic R5s110176 sic@

                                                                          f_GetAdditionalUpdateResult( v_ReceivedAsp.Signalling.Nas[0].Pdu.Msg.aTTACH_REQUEST.addUpdateType)))));

    //@siclog "Step 30" siclog@

    //The UE sends TRACKING AREA UPDATE COMPLETE

    SRB.receive ( car_SRB2_NasPdu_IND ( eutra_CellA,

                                        cr_NAS_Indication ( tsc_SHT_IntegrityProtected_Ciphered,

                                                            cr_508_TAU_Complete )));

    f_SendIRATCoOrd (GERAN, cms_IRAT_Trigger); //send Trigger to release GERAN cell.

  } // function fl_TC_9_2_3_4_1_Body


After change:

	  function fl_TC_9_2_3_4_1_Body() runs on EUTRA_PTC

  {

    var SRB_COMMON_IND v_ReceivedAsp;

    var template (omit) EPS_BearerContextStatus v_EpsBearerCtxtStatus;

    var IRAT_CoOrd_Auth_Type v_CoOrd_AuthParams;

    var EPS_UpdateTypeValue v_UpdateType := f_GetEPSTAUType(NORMAL); // @sic R5s110176 sic@

    var EPS_UpdateTypeValue v_UpdateInd := f_GetEPSTAUType(NORMAL);
    // ==========================================================================

    // initialize component specific data structures

    // ==========================================================================

    // get all data which are required for a GUTI

    var GutiParameters_Type v_Guti1_Params := f_EUTRA_CellInfo_GetGuti(eutra_CellA);

    var GutiParameters_Type v_Guti2_Params := f_CreateGutiParamsWithUpdatedSTmsi(v_Guti1_Params);

    var GutiParameters_Type v_Guti3_Params := f_CreateGutiParamsWithUpdatedSTmsi(v_Guti2_Params);

    var O10_Type v_GutiStringFromGERAN;

    var NAS_PlmnId v_PLMN := f_Asn2Nas_PlmnId(v_Guti1_Params.PLMN_Identity);

    var NAS_Tac v_TAC := bit2oct(f_EUTRA_CellInfo_GetTAC(eutra_CellA));

    var RRC_TransactionIdentifier v_RRC_TI := tsc_RRC_TI_Def;

    var NAS_Lac v_LAC := f_EUTRA_CellInfo_GetLocationAreaCode ( eutra_CellA );

    var template (omit) LocAreaId v_LAI := f_GetLAI ( v_PLMN, v_LAC, NORMAL); // @sic R5s110176 sic@

    var EUTRA_SecurityParams_Type v_SecurityParams := f_EUTRA_Security_Get();

    var Common_AuthenticationParams_Type v_Auth_Params;

    var template AdditionalUpdateType v_AdditionalUpdateType := f_GetAdditionalUpdateType ();

    var GutiParameters_Type v_Guti_CellB_Params := f_EUTRA_CellInfo_GetGuti(eutra_CellB);
    v_SecurityParams.AuthParams.KeySeq := v_SecurityParams.KSIasme; // The KSIasme value needs to be passed over as it will be used as the GERAN CKSN

    f_EUTRA_Security_Set(v_SecurityParams);

    v_Auth_Params := f_AuthenticationInit(v_SecurityParams.AuthParams);

    f_EUTRA_SendAuthCoOrdParameters(GERAN, eutra_CellA, v_Auth_Params);

    //+ Steps 1 to step 9 are implemented in GERAN side.

    //@siclog "Step 10 - 12 Void" siclog@

    //@siclog "Step 13" siclog@

    //+   The signal strength of Cell A is raised to that of the Serving Cell and that of Cell 24 is lowered to that of a Suitable Neighbour Cell as defined in table 6.2.2.1-1 of TS 36.508.

    //    Note: Cell 24 is still suitable but the UE is expected to select Cell A.

    v_CoOrd_AuthParams := f_ReceiveAuthCoOrdParameters(GERAN);

    f_EUTRA_SetCellPower(eutra_CellA, tsc_ServingCellRS_EPRE);

    v_GutiStringFromGERAN := v_PLMN & bit2oct(replace(v_CoOrd_AuthParams.GutiParams.MME_GroupId, 0, 1, '0'B)) & bit2oct(v_CoOrd_AuthParams.GutiParams.MME_Code) & bit2oct(v_CoOrd_AuthParams.GutiParams.M_TMSI);           //NOTE 1
    //@siclog "Step 14 Void" siclog@

    //@siclog "Step 15" siclog@

    //+   Check: does the UE send an RRCConnectionRequest with the InitialUE-Identity set to "randomValue" and the establishmentcause

    //     set to MO-signalling on Cell A?

    // Editor's Note: Cell reselection to E-UTRAN might not occur until the GPRS READY timer has expired. Hence this step might occur up to 40 seconds after step 13.

    f_EUTRA_RRC_ConnectionRequest_Def(eutra_CellA, cr_EstablishmentCause_MoSignalling, cr_InitialUE_Identity_NotSTMSI);   /* @sic R5-120721 change 3 sic@ */

    //@siclog "Step 16" siclog@

    //+   The SS responds with RRCConnectionSetup

    f_EUTRA_RRC_ConnectionSetup_Def ( eutra_CellA );

    //@siclog "Step 17" siclog@

    //+   Check: does the UE send an RRCConnectionSetupComplete with the mmegi and mmec set to the values derived from the mapped RAI and P-TMSI; [1,P]

    //  Check: are the contents of the TRACKING AREA UPDATE REQUEST with the correct parameters? [2,3,P]

    SRB.receive(car_SRB1_RrcNasPdu_IND(eutra_CellA,

                                       cr_508_RRCConnectionSetupComplete(v_RRC_TI, ?,

                                                                         cr_RegisteredMME(*,       //NOTE 1
                                                                                          replace(v_CoOrd_AuthParams.GutiParams.MME_GroupId, 0, 1, '0'B),
                                                                                          v_CoOrd_AuthParams.GutiParams.MME_Code)),

                                       cr_NAS_Indication(tsc_SHT_IntegrityProtected,

                                      cdr_TAU_Request_OldandAdditionalGuti_92341(v_UpdateType, //NOTE 2
                                                                                              f_EUTRA_SecurityKSIasme_Get(),

                                                                                              v_AdditionalUpdateType,

                                                                                              cs_MobileIdentityGuti(omit, v_GutiStringFromGERAN),

                                                                                              cr_CiphKeySeqNum ('8'H, v_CoOrd_AuthParams.AuthPS.KeySeq),

                                                                                              f_GutiParameters2MobileIdentity(tsc_IEI_Guti, v_Guti_CellB_Params), 
                                                                                              cr_UeRadioCap)))) -> value v_ReceivedAsp;

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 17");

    //@siclog "Step 18" siclog@

    //+   The SS sends TRACKING AREA UPDATE ACCEPT

    //  Note: the default message contents cause the allocation of a new GUTI and new TAI list

    v_EpsBearerCtxtStatus := v_ReceivedAsp.Signalling.Nas[0].Pdu.Msg.tRACKING_AREA_UPDATE_REQUEST.epsBearerContextStatus;

    SRB.send(cas_SRB1_NasPdu_REQ(eutra_CellA,

                                 cs_TimingInfo_Now,

                                 cs_NAS_Request(

                                                tsc_SHT_IntegrityProtected_Ciphered,

                                                cs_508_TAU_Accept(

                                                                  v_UpdateType,

                                                                  f_GutiParameters2MobileIdentity(tsc_IEI_Guti, v_Guti2_Params),

                                                                  cds_TAIListNonConsecutive_tlv(v_PLMN,

                                                                                                { '0001'O, '0002'O }), // TAC1=1, TAC2=2

                                                                  v_EpsBearerCtxtStatus,

                                                                  v_LAI,

                                                                  f_GetMSId (NORMAL), // @sic R5s110176 sic@

                                                                  f_GetAdditionalUpdateResult(v_ReceivedAsp.Signalling.Nas[0].Pdu.Msg.tRACKING_AREA_UPDATE_REQUEST.addUpdateType)) ) ) );

    //+  The UE sends TRACKING AREA UPDATE COMPLETE

    //@siclog "Step 19" siclog@

    SRB.receive(car_SRB1_NasPdu_IND(eutra_CellA,

                                    cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                      cr_508_TAU_Complete)));

    //+ the SS releases the RRC connection

    //@siclog "Step 20" siclog@

    f_EUTRA_RRC_ConnectionRelease(eutra_CellA);

    //+ Cell 24 is switched off

    //@siclog "Step 21" siclog@

    //@siclog "Step 22" siclog@

    // The signal strength of Cell 26 is raised to that of the Serving Cell and that of Cell A is lowered to that of a Suitable Neighbour Cell as defined in table 6.2.2.1-1 of TS 36.508.

    // Note: Cell A is still suitable but the UE is expected to select Cell 26.

    f_EUTRA_SetCellPower(eutra_CellA, tsc_NonSuitableCellRS_EPRE); //NOTE 3
    //f_SendIRATCoOrd (GERAN, cms_IRAT_Trigger);

    f_EUTRA_CellInfo_SetGuti(eutra_CellA, v_Guti2_Params); //NOTE 4
    v_SecurityParams := f_EUTRA_Security_Get();

    v_SecurityParams.AuthParams.KeySeq := v_SecurityParams.KSIasme;

    f_EUTRA_SendAuthCoOrdParameters(GERAN, eutra_CellA, v_SecurityParams.AuthParams); //NOTE 5
    //+ Step 21 to step 26 is implemented in GERAN side

    //@siclog "Step 27" siclog@

    //+  The signal strength of Cell A is raised to that of the Serving Cell and that of Cell 26 is lowered to that of a Suitable Neighbour Cell as defined in table 6.2.2.1-1 of TS 36.508.

    //   Note: Cell 26 is still suitable but the UE is expected to select Cell A.

    //f_WaitForIRATCoOrd_Trigger(GERAN);

    v_CoOrd_AuthParams := f_ReceiveAuthCoOrdParameters(GERAN); //NOTE 5 
    v_GutiStringFromGERAN := v_PLMN & bit2oct(replace(v_CoOrd_AuthParams.GutiParams.MME_GroupId, 0, 1, '0'B)) & bit2oct(v_CoOrd_AuthParams.GutiParams.MME_Code) & bit2oct(v_CoOrd_AuthParams.GutiParams.M_TMSI); //NOTE 6 
    f_EUTRA_SetCellPower(eutra_CellA, tsc_ServingCellRS_EPRE);

    //@siclog "Step 28" siclog@

    // Check: does the UE send on Cell A an RRCConnectionSetupComplete with the selectedPLMN-identity indicating the registered MME

    // (i.e. the PLMN in the GUTI allocated in step 18 ), and, the mmegi and mmec are set to the values in the GUTI allocated in step 18? [1,P]

    // Check: does the UE send a TRACKING AREA UPDATE REQUEST with the correct parameters? [2,6,P]

    f_EUTRA_RRC_ConnectionRequest_Def(eutra_CellA);    // @sic R5-120721 change 3 sic@

    f_EUTRA_RRC_ConnectionSetup_Def(eutra_CellA);

    if(v_UpdateInd == tsc_EpsUpdate_Combined_TaLaUpdate)  // @sic R5s110172 sic@ //NOTE 7
    {

      v_UpdateInd := tsc_EpsUpdate_Combined_IMSIAttach;

    }
    SRB.receive(car_SRB1_RrcNasPdu_IND(eutra_CellA,

                                       cr_508_RRCConnectionSetupComplete(v_RRC_TI, ?,

                                                                         cr_RegisteredMME(omit,                  //NOTE 6
                                                                                          replace(v_CoOrd_AuthParams.GutiParams.MME_GroupId, 0, 1, '0'B),
                                                                                          v_CoOrd_AuthParams.GutiParams.MME_Code)),

                                       cr_NAS_Indication(tsc_SHT_IntegrityProtected,

                                                     cdr_TAU_Request_OldandAdditionalGuti_92341(v_UpdateInd , //NOTE 1 and 7
                                                                                              f_EUTRA_SecurityKSIasme_Get(),

                                                                                              v_AdditionalUpdateType,

                                                                                              cs_MobileIdentityGuti(omit, v_GutiStringFromGERAN),

                                                                                              *,

                                                                                              f_GutiParameters2MobileIdentity(tsc_IEI_Guti, v_Guti2_Params),

                                                                                              omit)))) -> value v_ReceivedAsp;

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 28");

    //@siclog "Step 29" siclog@

    // The SS sends TRACKING AREA UPDATE ACCEPT

    // Note: the default message contents cause the allocation of a new GUTI.

    v_EpsBearerCtxtStatus := v_ReceivedAsp.Signalling.Nas[0].Pdu.Msg.tRACKING_AREA_UPDATE_REQUEST.epsBearerContextStatus;

    SRB.send ( cas_SRB1_NasPdu_REQ ( eutra_CellA,

                                     cs_TimingInfo_Now,

                                     cs_NAS_Request ( tsc_SHT_IntegrityProtected_Ciphered,

                                                      cs_508_TAU_Accept ( v_UpdateType,

                                                                          f_GutiParameters2MobileIdentity(tsc_IEI_Guti, v_Guti3_Params),

                                                                          cds_TAIListNonConsecutive_tlv( v_PLMN, {v_TAC } ),

                                                                          v_EpsBearerCtxtStatus,

                                                                          v_LAI,

                                                                          f_GetMSId (NORMAL), // @sic R5s110176 sic@

                                                                          f_GetAdditionalUpdateResult( v_ReceivedAsp.Signalling.Nas[0].Pdu.Msg.aTTACH_REQUEST.addUpdateType)))));

    //@siclog "Step 30" siclog@

    //The UE sends TRACKING AREA UPDATE COMPLETE

    SRB.receive ( car_SRB1_NasPdu_IND ( eutra_CellA, //NOTE 8
                                        cr_NAS_Indication ( tsc_SHT_IntegrityProtected_Ciphered,

                                                            cr_508_TAU_Complete )));

    f_EUTRA_CellInfo_SetGuti(eutra_CellA, v_Guti3_Params); //NOTE 9
    f_SendIRATCoOrd (GERAN, cms_IRAT_Trigger); //send Trigger to release GERAN cell.

  } // function fl_TC_9_2_3_4_1_Body


New Template:

	template NAS_UL_Message_Type cdr_TAU_Request_OldandAdditionalGuti_92341(EPS_UpdateTypeValue p_EpsUpdate_TypeValue,

                                                                    NAS_KsiValue p_KsiValue,

                                                                    template AdditionalUpdateType p_AdditionalUpdateType,

                                                                    template (present) MobileIdentity p_OldGuti,

                                                                    template CiphKeySeqNum p_CiphKeySeqNum,

                                                                    template (present) MobileIdentity p_AdditionalGuti,

                                                                    template (omit) UERadioCapInfoUpdateNeeded  p_UeRadioCap)

    modifies cr_508_TAU_Request :=

  {

    tRACKING_AREA_UPDATE_REQUEST := {

      asmeNasKeySetId            := cr_NAS_KeySetIdentifier (p_KsiValue, tsc_NasKsi_NativeSecurityContext),

      epsUpdateType              := cr_EPS_UpdateType(tsc_EpsUpdate_NotActive, p_EpsUpdate_TypeValue),

      oldGuti                    := p_OldGuti,

      gprsCipheringKeySeqNum     := p_CiphKeySeqNum,

      oldPtmsiSignature          := cr_PTMSI_SignatureAny,

      additionalGuti             := p_AdditionalGuti,

      nonce                      := cr_NonceUE_Any,

      drxParameter               := omit,

      ueRadioCapNeeded           := p_UeRadioCap

    }

  };


4.4 Change 3

	Testcase name
	9.2.3.4.1

	Reason for change
	1.  Accroding to 36.523-1, cells 24 and 26 are configured with the same MCC and MNC as cell A

2.  As per test purpose PDP context on GERAN should be active.

3.  At step 27 M- TMSI & Gutiparams will be derived as per the spec 3GPP TS 36.523-1(9.2.3.4.1) however TTCN doesn’t contain this (P-TMSI allocated  in step 8 and RAI allocated in step 25) (change 1.8).

4. At step 27 f_SendIRATCoOrd wont be used to send AuthParams.( f_GERAN_SendAuthParameters will be used instead of SendIRATCoOrd).

5. At step 24, UE will send the Routing Area Update request in this message addOldRAI should be derived from cell 24 and addMobileId should be TLV type.

6. At step 23, P-TMSI should be genereate from M_TMSI and MME_Code received in auth parameter received in IRAT coordination message.

7. At step 27, SS should send the authentication parameter to EUTRA PTC and the GUTI parameter for this should be generate from LAC, px_PTMSI_Def and TMSI Out.

Note: Prose CR will be raised for these changes at the next RAN5 meeting

	Summary of change
	1.   Configure the cell 24 and cell 26 on default PLMN.

2.   Added the code for PDP Activation after step 9.

3.   At step 27 M- TMSI & Gutiparams is derived as per the spec 3GPP TS 36.523-1(9.2.3.4.1)

4.   At step 27 SendAuthparams is used instead of SendIRATCoord

5.  Derive the addOldRAI from cell 24 and update addMobileId to TLV type.

6. Use M_TMSI and MME_Code receive in IRAT coordination message to derive P-TMSI at step 23.

7. Send authentication parameter to EUTAN PTC at step 23.

	Source of change
	NAS_TrackingArea_UG_GERAN.ttcn


Before change:

	  function f_TC_9_2_3_4_1_GERAN () runs on GERAN_PTC

  {

    var IRAT_CoOrd_Auth_Type v_CoOrd_AuthParams;

    var octetstring v_PLMN := f_U_Asn2Nas_PlmnId({ mcc := {9, 8, 7},  mnc := {6, 5} });
    var template (value) CoOrd_GUTIParam_Type v_GTUIParams;

    var G_LLC_DATAMESSAGE_IND v_LLCDataInd;

    var B3_Type v_UpdateType;

    var B32_Type v_TLLIBits;

    var O1_Type v_RAC;

    var O2_Type v_LAC;

    var B32_Type v_M_TMSIIn, v_M_TMSIOut;

    f_GERAN_Init(EUTRA_GERAN);

    v_RAC := f_GERAN_RAC_Get(geran_Cell26);

    v_LAC := f_GERAN_LAC_Get(geran_Cell26);

    f_GERAN_PLMN_Set(geran_Cell24, v_PLMN);

    f_GERAN_PLMN_Set(geran_Cell26, v_PLMN);

    f_GERAN_CreateCell (geran_Cell24);

    f_GERAN_CreateCell (geran_Cell26);

    f_GERAN_InterRAT_InitialiseAuthParams (); // Get and store Auth Params from EUTRA

    f_GERAN_TestBody_Set( true );

    // First Trigger

    //@siclog "Step 1" siclog@

    v_CoOrd_AuthParams := f_ReceiveAuthCoOrdParameters (EUTRA);

    v_M_TMSIIn := v_CoOrd_AuthParams.GutiParams.M_TMSI;

    f_GERAN_SetCellPowerLevel (geran_Cell24, tsc_GERAN_ServingCell);

    //+   The UE is switched on.

    f_UT_SwitchOnUE(UT, false); /* no USIM insertion */

    // The UE transmits an ATTACH REQUEST message on Cell 24

    // Step 3

    //The SS transmits an AUTHENTICATION REQUEST message.

    // Step 4

    //The UE transmits an AUTHENTICATION RESPONSE message.

    // Step 8

    // SS responds with ATTACH ACCEPT message including valid TMSI, P-TMSI (TIN set to P-TMSI) and RAI.

    //  Editor's Note: the default message contents in 51.010 clause 40.2.4.3 allocate a Negotiated READY timer value of 32 seconds.

    // Step 9

    // The UE transmits an ATTACH COMPLETE message.

    //@siclog "Step 3 - 9" siclog@

    f_GERAN_GPRSAttach (geran_Cell24);

    // Need to Set GUTI based on PTMSI (which is px_PTMSI_Def), LAC and RAC, see 23.003

    v_M_TMSIOut := '11'B & substr(oct2bit(px_PTMSI_Def), 24, 6) & oct2bit(v_RAC) & substr(oct2bit(px_PTMSI_Def), 0, 15);
    v_GTUIParams := cs_IRAT_GUTIStuff (replace(oct2bit(v_LAC), 0, 1, '1'B), substr(oct2bit(px_PTMSI_Def), 16, 8), v_M_TMSIOut);
    f_GERAN_SetCellPowerLevel (geran_Cell24, tsc_GERAN_SuitableNeighbourCell);

    f_GERAN_SendAuthParameters (EUTRA, v_GTUIParams);

    // Step 10 to step 20 is implementated in EUTRA side.

    // Second Trigger:

    //  Cell 24 is switched off

    //@siclog "Step 21" siclog@

    f_WaitForIRATCoOrd_Trigger (EUTRA);
    f_GERAN_SetCellPowerLevel (geran_Cell24, tsc_GERAN_ChPwrLvl_Off);

    //@siclog "Step 22" siclog@

    // The signal strength of Cell 26 is raised to that of the Serving Cell and that of Cell A is lowered to that of a Suitable Neighbour Cell as defined in table 6.2.2.1-1 of TS 36.508 [18].

    // Note: Cell A is still suitable but the UE is expected to select Cell 26.

     f_GERAN_SetCellPowerLevel (geran_Cell26, tsc_GERAN_ServingCell);

    //+ Check: does the UE send on Cell 26 a ROUTING AREA UPDATE REQUEST with the correct parameters? [5,P]

    //@siclog "Step 24" siclog@

    v_LLCDataInd := f_GPRS_ReceiveUplinkMsg (geran_Cell26, car_G_LLC_UnitData_IND ( cr_G_RAUReq_Ids (cr_GMM_UpdateType, cr_RAI (omit), cr_PTMSI_SignatureAny ifpresent,

                                        v_CoOrd_AuthParams.AuthPS.KeySeq,  // Get value from Authentication done on EUTRA during Preamble

                                        omit,

                                        cs_MobileIdTMSIorPTMSI ('18'O, px_PTMSI_Def),

                                        cr_RoutingAreaId2(v_PLMN&v_LAC&v_RAC))));

    v_UpdateType := f_UG_RauReq_GetUpdateValueForRauAcc(v_LLCDataInd.unit.msg.rauReq);
    //+Check: does the UE send on Cell 26 an RLC/MAC header with the TLLI derived from the P-TMSI that is derived from the GUTI allocated in step 18? [4,P]

    // The RLC/MAC header will be sent and the TLLI changed during f_GPRS_ReceiveUplinkMsg, but we can check it now

    //@siclog "Step 23" siclog@

    v_TLLIBits := oct2bit(f_GERAN_TLLI_Get(geran_Cell26));

    if ((substr(v_M_TMSIIn, 24, 6) == substr(v_TLLIBits, 24, 6)) and (substr(v_M_TMSIIn, 0, 16) == substr(v_TLLIBits, 0, 16)))
    {

       f_GERAN_PreliminaryPass(__FILE__, __LINE__, "Step 23");

    } else {

       f_GERAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "PTMSI doesn't correlate to M-TMSI, Step 23");

    }

    //+The SS sends ROUTING AREA UPDATE ACCEPT (without the allocated P-TMSI IE, but with the Requested MS Information IE indicating "E-UTRAN inter RAT information container IE requested").

    //@siclog "Step 25" siclog@

    f_GPRS_SendDownlinkMsg(geran_Cell26, 1, cas_G_LLC_UnitData_Req (bit2oct(v_TLLIBits), tsc_LLCSapi_GMM, tsc_LLC_PM, px_GSM_CipheringOnOff,

                  cs_G_RA_UpdAcc3 (cs_GMM_UpdateResult(v_UpdateType), cs_RAI(omit, cs_LAI(omit, v_PLMN, v_LAC), v_RAC),

                                cs_PTMSI_Signature (px_PTMSI_SigDef), omit, cs_RequestedMS_Info)));

    //+ UE sends ROUTING AREA UPDATE COMPLETE message

    //@siclog "Step 26" siclog@

    v_LLCDataInd := f_GPRS_ReceiveUplinkMsg (geran_Cell26, car_G_LLC_UnitData_IND (cdr_G_RA_UpdCompleteWithIRInfo));

    //+  The signal strength of Cell A is raised to that of the Serving Cell and that of Cell 26 is lowered to that of a Suitable Neighbour Cell as defined in table 6.2.2.1-1 of TS 36.508 [18].

    //   Note: Cell 26 is still suitable but the UE is expected to select Cell A.

    //@siclog "Step 27" siclog@

    f_GERAN_SetCellPowerLevel (geran_Cell26, tsc_GERAN_SuitableNeighbourCell);

    // Finished here for now

    f_SendIRATCoOrd (EUTRA, cms_IRAT_Trigger);
    f_WaitForIRATCoOrd_Trigger(EUTRA);  // wait for E-UTRAN side to finish and then the whole test case is finished.

    // Test case now finished on GERAN side

    f_GERAN_TestBody_Set( false );

    f_GERAN_SS_CellRelease ( geran_Cell24 );

    f_GERAN_SS_CellRelease ( geran_Cell26 );

  } // f_TC_9_2_3_4_1_GERAN


After change:

	    import from GERAN_CellInfo all; //NOTE 1
function f_TC_9_2_3_4_1_GERAN () runs on GERAN_PTC

  {

    var IRAT_CoOrd_Auth_Type v_CoOrd_AuthParams;

    var octetstring v_PLMN := f_U_Asn2Nas_PlmnId({ mcc := {0, 0, 1},  mnc := {0, 1} }); //NOTE 1
    var template (value) CoOrd_GUTIParam_Type v_GTUIParams;

    var G_LLC_DATAMESSAGE_IND v_LLCDataInd;

    var B3_Type v_UpdateType;

    var B32_Type v_TLLIBits;

    var O1_Type v_RAC, v_RAC_Cell24; //NOTE 3
    var O2_Type v_LAC, v_LAC_Cell24;
    var B32_Type v_M_TMSIIn, v_M_TMSIOut;

    var GERAN_SystemInformation_Type v_SysInfo;

    var bitstring v_PTMSI;
    f_GERAN_Init(EUTRA_GERAN);

    v_RAC := f_GERAN_RAC_Get(geran_Cell26);

    v_LAC := f_GERAN_LAC_Get(geran_Cell26);

    v_RAC_Cell24 := f_GERAN_RAC_Get(geran_Cell24); //NOTE 3
    v_LAC_Cell24 := f_GERAN_LAC_Get(geran_Cell24);
    f_GERAN_CreateCell (geran_Cell24);

    f_GERAN_CreateCell (geran_Cell26);

    f_GERAN_InterRAT_InitialiseAuthParams (); // Get and store Auth Params from EUTRA

    f_GERAN_TestBody_Set( true );

    // First Trigger

    //@siclog "Step 1" siclog@

    v_CoOrd_AuthParams := f_ReceiveAuthCoOrdParameters (EUTRA);

    v_M_TMSIIn := v_CoOrd_AuthParams.GutiParams.M_TMSI;

    f_GERAN_SetCellPowerLevel (geran_Cell24, tsc_GERAN_ServingCell);

    //+   The UE is switched on.

    f_UT_SwitchOnUE(UT, false); /* no USIM insertion */

    // The UE transmits an ATTACH REQUEST message on Cell 24

    // Step 3

    //The SS transmits an AUTHENTICATION REQUEST message.

    // Step 4

    //The UE transmits an AUTHENTICATION RESPONSE message.

    // Step 8

    // SS responds with ATTACH ACCEPT message including valid TMSI, P-TMSI (TIN set to P-TMSI) and RAI.

    //  Editor's Note: the default message contents in 51.010 clause 40.2.4.3 allocate a Negotiated READY timer value of 32 seconds.

    // Step 9

    // The UE transmits an ATTACH COMPLETE message.

    //@siclog "Step 3 - 9" siclog@

    f_GERAN_GPRSAttach (geran_Cell24);

    f_GERAN_ActivatePDPContext (geran_Cell24); //NOTE 2
    f_GERAN_LLC_XID (geran_Cell24, true);
    // Need to Set GUTI based on PTMSI (which is px_PTMSI_Def), LAC and RAC, see 23.003

    v_M_TMSIOut := '11'B & substr(oct2bit(px_PTMSI_Def), 2, 6) & oct2bit(v_RAC_Cell24) & substr(oct2bit(px_PTMSI_Def), 16, 16); //NOTE 3
    v_GTUIParams := cs_IRAT_GUTIStuff (replace(oct2bit(v_LAC_Cell24), 0, 1, '1'B), substr(oct2bit(px_PTMSI_Def), 8, 8), v_M_TMSIOut); //NOTE 3
    f_GERAN_SetCellPowerLevel (geran_Cell24, tsc_GERAN_SuitableNeighbourCell);

    f_GERAN_SendAuthParameters (EUTRA, v_GTUIParams);

    // Step 10 to step 20 is implementated in EUTRA side.

    // Second Trigger:

    //  Cell 24 is switched off

    //@siclog "Step 21" siclog@

    v_CoOrd_AuthParams := f_ReceiveAuthCoOrdParameters (EUTRA);    //NOTE 4
    f_GERAN_SetCellPowerLevel (geran_Cell24, tsc_GERAN_ChPwrLvl_Off);

    //@siclog "Step 22" siclog@

    // The signal strength of Cell 26 is raised to that of the Serving Cell and that of Cell A is lowered to that of a Suitable Neighbour Cell as defined in table 6.2.2.1-1 of TS 36.508 [18].

    // Note: Cell A is still suitable but the UE is expected to select Cell 26.

     f_GERAN_SetCellPowerLevel (geran_Cell26, tsc_GERAN_ServingCell);

    //+ Check: does the UE send on Cell 26 a ROUTING AREA UPDATE REQUEST with the correct parameters? [5,P]

    //@siclog "Step 24" siclog@

    v_LLCDataInd := f_GPRS_ReceiveUplinkMsg (geran_Cell26, car_G_LLC_UnitData_IND ( cr_G_RAUReq_Ids (cr_GMM_UpdateType, cr_RAI (omit), cr_PTMSI_SignatureAny ifpresent,

                                        v_CoOrd_AuthParams.AuthPS.KeySeq,  // Get value from Authentication done on EUTRA during Preamble

                                        omit,

                                        cs_MobileIdTMSIorPTMSI ('1A'O, px_PTMSI_Def), //NOTE 5
                                        cr_RoutingAreaId2(v_PLMN&v_LAC_Cell24&v_RAC_Cell24))));

    v_UpdateType := v_LLCDataInd.unit.msg.rauReq.updateType.updateValue;
    if (v_UpdateType != '000'B){

      v_UpdateType := '001'B;

    }

    //+Check: does the UE send on Cell 26 an RLC/MAC header with the TLLI derived from the P-TMSI that is derived from the GUTI allocated in step 18? [4,P]

    // The RLC/MAC header will be sent and the TLLI changed during f_GPRS_ReceiveUplinkMsg, but we can check it now

    //@siclog "Step 23" siclog@

    v_PTMSI := '11'B & substr (v_CoOrd_AuthParams.GutiParams.M_TMSI, 2, 6) & v_CoOrd_AuthParams.GutiParams.MME_Code &  substr (v_CoOrd_AuthParams.GutiParams.M_TMSI, 16, 16); //NOTE 6
    v_TLLIBits := oct2bit(f_GERAN_TLLI_Get(geran_Cell26));

    if (substr(v_TLLIBits, 2, 30) == substr(v_PTMSI, 2, 30)) //NOTE 6
    {

       f_GERAN_PreliminaryPass(__FILE__, __LINE__, "Step 23");

    } else {

       f_GERAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "PTMSI doesn't correlate to M-TMSI, Step 23");

    }

    //+The SS sends ROUTING AREA UPDATE ACCEPT (without the allocated P-TMSI IE, but with the Requested MS Information IE indicating "E-UTRAN inter RAT information container IE requested").

    //@siclog "Step 25" siclog@

    f_GPRS_SendDownlinkMsg(geran_Cell26, 1, cas_G_LLC_UnitData_Req (bit2oct(v_TLLIBits), tsc_LLCSapi_GMM, tsc_LLC_PM, px_GSM_CipheringOnOff,

                  cs_G_RA_UpdAcc3 (cs_GMM_UpdateResult(v_UpdateType), cs_RAI(omit, cs_LAI(omit, v_PLMN, v_LAC), v_RAC),

                                cs_PTMSI_Signature (px_PTMSI_SigDef), omit, cs_RequestedMS_Info)));

    //+ UE sends ROUTING AREA UPDATE COMPLETE message

    //@siclog "Step 26" siclog@

    v_LLCDataInd := f_GPRS_ReceiveUplinkMsg (geran_Cell26, car_G_LLC_UnitData_IND (cdr_G_RA_UpdCompleteWithIRInfo));

    //+  The signal strength of Cell A is raised to that of the Serving Cell and that of Cell 26 is lowered to that of a Suitable Neighbour Cell as defined in table 6.2.2.1-1 of TS 36.508 [18].

    //   Note: Cell 26 is still suitable but the UE is expected to select Cell A.

    //@siclog "Step 27" siclog@

    v_M_TMSIOut    := '11'B & substr(oct2bit(px_PTMSI_Def), 2, 6) & oct2bit(v_RAC) & substr(oct2bit(px_PTMSI_Def), 16, 16);

    v_GTUIParams := cs_IRAT_GUTIStuff (oct2bit(v_LAC), substr(oct2bit(px_PTMSI_Def), 8, 8), v_M_TMSIOut); //NOTE 7
    f_GERAN_SetCellPowerLevel (geran_Cell26, tsc_GERAN_SuitableNeighbourCell);

    // Finished here for now

    f_GERAN_SendAuthParameters (EUTRA, v_GTUIParams); //NOTE 7
    f_WaitForIRATCoOrd_Trigger(EUTRA);  // wait for E-UTRAN side to finish and then the whole test case is finished.

    // Test case now finished on GERAN side

    f_GERAN_TestBody_Set( false );

    f_GERAN_SS_CellRelease ( geran_Cell24 );

    f_GERAN_SS_CellRelease ( geran_Cell26 );

  } // f_TC_9_2_3_4_1_GERAN


4.5 Change 4

	Testcase name
	9.2.3.4.1

	Reason for change
	1. As per 24.008 clauses 9.5.2, Negotiated QoS should be the LV type.

2. For UE supporting GERAN, Test case should send with LLC SAPI set as 3.

	Summary of change
	1. Encode Negotiated QoS as LV insted of TLV.

2. Use template “cs_LLC_SAPI_03”.

	Source of change
	GERAN_CommonFunctions.ttcn


Before change:

	  function f_GERAN_ActivatePDPContext(GERAN_CellId_Type p_CellId,

                                      template (value) QualityOfService p_QoS := cs_QoS_PDP2) runs on GERAN_PTC

  {

    var G_LLC_DATAMESSAGE_IND v_LLCDataInd;

    var TI v_TI;

    var template (value) PktDataProtoAddr v_PDPAddr1 := f_UG_GetPktDataProtoAddr(true);

    f_UT_ActivatePDPContextUG (UT,

                               1, // CId, don't care what the value is

                               p_QoS);

    v_LLCDataInd := f_GPRS_ReceiveUplinkMsg (p_CellId, car_G_LLC_UnitData_IND ( cr_G_ActPDP_ContextReqMO_Any));

    v_TI := v_LLCDataInd.unit.msg.actPDPContextReq.ti;

    // Send Accept

    v_TI.tiFlag := '1'B;

    f_GPRS_SendDownlinkMsg(p_CellId, 1, cas_G_LLC_UnitData_Req (f_GERAN_TLLI_Get(p_CellId),

                                                                tsc_LLCSapi_GMM,

                                                                tsc_LLC_PM,

                                                                px_GSM_CipheringOnOff,

                                                                cs_G_PDPContextAccept (v_TI,

                                                                                       cs_LLC_SAPI_11,

                                                                                       p_QoS,

                                                                                       v_PDPAddr1,

                                                                                       cs_ProtocolConfigOptions_PPP)));

  }// end of f_GERAN_ActivatePDPContext


After change:

	  function f_GERAN_ActivatePDPContext(GERAN_CellId_Type p_CellId,

                                      template (value) QualityOfService p_QoS := cs_QoS_PDP4) runs on GERAN_PTC

  {

    var G_LLC_DATAMESSAGE_IND v_LLCDataInd;

    var TI v_TI;

    var template (value) PktDataProtoAddr v_PDPAddr1 := f_UG_GetPktDataProtoAddr(true);

    f_UT_ActivatePDPContextUG (UT,

                               1, // CId, don't care what the value is

                               p_QoS);

    v_LLCDataInd := f_GPRS_ReceiveUplinkMsg (p_CellId, car_G_LLC_UnitData_IND ( cr_G_ActPDP_ContextReqMO_Any));

    v_TI := v_LLCDataInd.unit.msg.actPDPContextReq.ti;

    // Send Accept

    v_TI.tiFlag := '1'B;

    f_GPRS_SendDownlinkMsg(p_CellId, 1, cas_G_LLC_UnitData_Req (f_GERAN_TLLI_Get(p_CellId),

                                                                tsc_LLCSapi_GMM,

                                                                tsc_LLC_PM,

                                                                px_GSM_CipheringOnOff,

                                                                cs_G_PDPContextAccept (v_TI,

                                                                                       cs_LLC_SAPI_03, 
                                                                                       p_QoS,

                                                                                       v_PDPAddr1,

                                                                                       cs_ProtocolConfigOptions_PPP)));

  }// end of f_GERAN_ActivatePDPContext


New Template:

	  template (value) QualityOfService cs_QoS_PDP4 modifies cs_QoS_PDP2 :=

  { /* @status    APPROVED */

    iei := omit

  };  


5 Execution Log Files

5.1 Nvidia Icera 410

The Nvidia Icera 410 UE passed this test case on the Anite Conformance Toolset solution. The documentation below is enclosed as evidence of the successful test case run [1]:

· Test Case Execution log files TC_9_2_3_2_1c_ICERA.htm:

In the log files (in html format) the complete test case execution can be seen. All message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log files.

6. References

	[1]
	R5s120219: This archive comprises html format execution log file.

